# $Id: PKGBUILD 88096 2013-04-11 13:54:05Z mtorromeo $
# Maintainer: Massimiliano Torromeo <massimiliano.torromeo@gmail.com>
# Contributor: Kovivchak Evgen <oneonfire@gmail.com>

pkgname=jemalloc
pkgver=3.3.1
pkgrel=1
pkgdesc="General-purpose scalable concurrent malloc implementation"
arch=('i686' 'x86_64')
license=('BSD')
url="http://www.canonware.com/jemalloc/"
depends=('glibc')
makedepends=('autoconf' 'make')
source=(http://www.canonware.com/download/jemalloc/$pkgname-$pkgver.tar.bz2)

build() {
	cd "$srcdir/$pkgname-$pkgver"
	./configure --prefix=/usr
	make
}

package() {
	cd "$srcdir/$pkgname-$pkgver"
	make DESTDIR="$pkgdir" install
	mv "$pkgdir"/usr/bin/{,jemalloc-}pprof
	chmod 644 "$pkgdir"/usr/lib/*.a
	install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/COPYING"
}

sha256sums=('35c433cc1df5cdf9eb58b7980338552fc1d7aa64f89fe5643a972ccedb7cf20a')
